Problem Chapter I.9

The planar R-RTR mechanism considered is shown in Fig. 1. The driver link is the rigid link 1 (the link AB). The following numerical data are given: $AB = 0.10$ m, $AC = 0.05$ m, and $CD = 0.15$ m. The angle of the driver link 1 with the horizontal axis is $\phi = 30°$. The constant angular speed of the driver link 1 is -50 rpm.

Position analysis for an input angle

Position of joint A

A Cartesian reference frame xOy is selected. The joint A is in the origin of the reference frame, that is, $A \equiv O$,

$$x_A = 0 \quad \text{and} \quad y_A = 0. \tag{1}$$

Position of joint C

The coordinates of the joint C are

$$x_C = AC = 0.05 \quad \text{and} \quad y_C = 0 \text{ m}. \tag{2}$$

Position of joint B

The unknowns are the coordinates of the joint B, x_B and y_B. Because the joint A is fixed and the angle ϕ is known, the coordinates of the joint B are computed from the following expressions

$$x_B = AB \cos\phi = 0.10 \cos 30° = 0.086 \text{ m},$$
$$y_B = AB \sin\phi = 0.10 \sin 30° = 0.050 \text{ m}. \tag{3}$$

Angle ϕ_2

The angle of link 2 (or link 3) with the horizontal axis is calculated from the slope of the straight line BC:

$$\phi_2 = \phi_3 = \arctan\frac{y_B - y_C}{x_B - x_C} = \arctan\frac{0.050}{0.086 - 0.050} = 0.938 \text{ rad} = 53.794°.$$

Position of joint D

The unknowns are the coordinates of the joint D, x_D and y_D

$$x_D = x_C + CD\cos\phi_3 = 0.050 + 0.15\cos 53.794° = 0.138 \text{ m},$$
$$y_D = y_C + CD\sin\phi_3 = 0.15 \sin 53.794° = 0.121 \text{ m}. \tag{4}$$

Velocity and Acceleration Analysis

Algebraic Method

The velocity of the point $B = B_1$ on the link 1 is

$$\mathbf{v}_B = \mathbf{v}_{B_1} = \mathbf{v}_A + \boldsymbol{\omega}_1 \times \mathbf{r}_{AB} = \boldsymbol{\omega}_1 \times \mathbf{r}_B,$$

where $\mathbf{v}_A \equiv \mathbf{0}$ is the velocity of the origin $A \equiv O$.
The angular velocity of link 1 is

$$\boldsymbol{\omega}_1 = \omega_1 \mathbf{k} = \frac{\pi n}{30}\mathbf{k} = \frac{\pi(-50)}{30}\mathbf{k} = -5.235\mathbf{k} \ \ \text{rad/s}.$$

the position vector of point B is

$$\mathbf{r}_{AB} = \mathbf{r}_B - \mathbf{r}_A = \mathbf{r}_B = x_B\boldsymbol{\imath} + y_B\boldsymbol{\jmath} + z_B\mathbf{k} = 0.086\boldsymbol{\imath} + 0.050\boldsymbol{\jmath} \ \ \text{m}.$$

The velocity of point B_2 on the link 2 is $\mathbf{v}_{B_2} = \mathbf{v}_{B_1}$ because between the links 1 and 2 there is a rotational joint.
The velocity of $B_1 = B_2$ is

$$\mathbf{v}_{B_1} = \mathbf{v}_{B_2} = \begin{vmatrix} \boldsymbol{\imath} & \boldsymbol{\jmath} & \mathbf{k} \\ 0 & 0 & \omega \\ x_B & y_B & 0 \end{vmatrix} = \begin{vmatrix} \boldsymbol{\imath} & \boldsymbol{\jmath} & \mathbf{k} \\ 0 & 0 & -5.235 \\ 0.086 & 0.050 & 0 \end{vmatrix} = 0.261\boldsymbol{\imath} - 0.453\boldsymbol{\jmath} \ \ \text{m/s}.$$

The acceleration of the point $B = B_1$ on the link 1 is

$$\begin{aligned} \mathbf{a}_B &= \mathbf{a}_{B_1} = \mathbf{a}_{B_2} = \mathbf{a}_A + \boldsymbol{\alpha}_1 \times \mathbf{r}_B + \boldsymbol{\omega}_1 \times (\boldsymbol{\omega}_1 \times \mathbf{r}_B) = \boldsymbol{\alpha}_1 \times \mathbf{r}_B - \boldsymbol{\omega}_1^2 \mathbf{r}_B \\ &= -\boldsymbol{\omega}_1^2 \mathbf{r}_B = -(-5.235)^2(0.086\boldsymbol{\imath} + 0.050\boldsymbol{\jmath}) = -2.374\boldsymbol{\imath} - 1.370\boldsymbol{\jmath} \ \ \text{m/s}^2. \end{aligned}$$

The angular acceleration of link 1 is $\boldsymbol{\alpha}_1 = \dot{\boldsymbol{\omega}}_1 = \mathbf{0}$.
The velocity of the point B_3 on the link 3 is calculated in terms of the velocity of the point B_2 on the link 2

$$\mathbf{v}_{B_3} = \mathbf{v}_{B_2} + \mathbf{v}_{B_3B_2}^{rel} = \mathbf{v}_{B_2} + \mathbf{v}_{B_{32}}, \tag{5}$$

where $\mathbf{v}_{B_3B_2}^{rel} = \mathbf{v}_{B_{32}}$ is the relative acceleration of B_3 with respect to B_2 on link 3. This relative velocity is parallel to the sliding direction BC, $\mathbf{v}_{B_{32}} || BC$, or

$$\mathbf{v}_{B_{32}} = v_{B_{32}} \cos\phi_2 \boldsymbol{\imath} + v_{B_{32}} \sin\phi_2 \boldsymbol{\jmath}, \tag{6}$$

where $\phi_2 = 53.794°$ is known from position analysis. The points B_3 and C are on the link 3 and

$$\mathbf{v}_{B_3} = \mathbf{v}_C + \boldsymbol{\omega}_3 \times \mathbf{r}_{CB} = \boldsymbol{\omega}_3 \times (\mathbf{r}_B - \mathbf{r}_C), \tag{7}$$

where $\mathbf{v}_C \equiv \mathbf{0}$ and the angular velocity of link 3 is

$$\boldsymbol{\omega}_3 = \omega_3 \mathbf{k}.$$

Equations (5), (6), and (7) give

$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \omega_3 \\ x_B - x_C & y_B - y_C & 0 \end{vmatrix} = \mathbf{v}_{B_2} + v_{B_{32}} \cos\phi_2 \mathbf{\imath} + v_{B_{32}} \sin\phi_2 \mathbf{\jmath}. \tag{8}$$

Equation (8) represents a vectorial equations with two scalar components on x-axis and y-axis and with two unknowns ω_3 and $v_{B_{32}}$

$$-\omega_3(y_B - y_C) = v_{Bx} + v_{B_{32}} \cos\phi_2,$$
$$\omega_3(x_B - x_C) = v_{By} + v_{B_{32}} \sin\phi_2,$$

or

$$-\omega_3(0.050) = 0.261 + v_{B_{32}} \cos 53.794°,$$
$$\omega_3(0.086 - 0.05) = -0.453 + v_{B_{32}} \sin 53.794°.$$

It results

$$\omega_3 = \omega_2 = -7.731 \text{ rad/s} \quad \text{and} \quad v_{B_{32}} = 0.211 \ \ \text{m/s}.$$

The acceleration of the point B_3 on the link 3 is calculated in terms of the acceleration of the point B_2 on the link 2

$$\mathbf{a}_{B_3} = \mathbf{a}_{B_2} + \mathbf{a}^{rel}_{B_3B_2} + \mathbf{a}^{cor}_{B_3B_2} = \mathbf{a}_{B_2} + \mathbf{a}_{B_{32}} + \mathbf{a}^{cor}_{B_{32}}, \tag{9}$$

where $\mathbf{a}^{rel}_{B_3B_2} = \mathbf{a}_{B_{32}}$ is the relative acceleration of B_3 with respect to B_2 on link 3. This relative acceleration is parallel to the sliding direction BC, $\mathbf{a}_{B_{32}} || BC$, or

$$\mathbf{a}_{B_{32}} = a_{B_{32}} \cos\phi_2 \mathbf{\imath} + a_{B_{32}} \sin\phi_2 \mathbf{\jmath}. \tag{10}$$

The Coriolis acceleration of B_3 realative to B_2 is

$$\mathbf{a}^{cor}_{B_{32}} = 2\ \boldsymbol{\omega}_3 \times \mathbf{v}_{B_{32}} = 2\ \boldsymbol{\omega}_2 \times \mathbf{v}_{B_{32}} = 2 \begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \omega_3 \\ v_{B_{32}} \cos\phi_2 & v_{B_{32}} \sin\phi_2 & 0 \end{vmatrix} =$$
$$2(-\omega_3 v_{B_{32}} \sin\phi_2 \mathbf{\imath} + \omega_3 v_{B_{32}} \cos\phi_2 \mathbf{\jmath}) =$$
$$2[-(-7.731)(0.211)\sin 53.794°\mathbf{\imath} + (-7.731)(0.211)\cos 53.794°\mathbf{\jmath}] =$$
$$2.635\mathbf{\imath} - 1.929\mathbf{\jmath}\ \ \text{m/s}^2. \tag{11}$$

The points B_3 and C are on the link 3 and

$$\mathbf{a}_{B_3} = \mathbf{a}_C + \boldsymbol{\alpha}_3 \times \mathbf{r}_{CB} - \omega_3^2 \mathbf{r}_{CB}, \tag{12}$$

where $\mathbf{a}_C \equiv \mathbf{0}$ and the angular acceleration of link 3 is

$$\boldsymbol{\alpha}_3 = \alpha_3 \mathbf{k}.$$

Equations (9), (10), (11), and (12) give

$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \alpha_3 \\ x_B - x_C & y_B - y_C & 0 \end{vmatrix} - \omega_3^2(\mathbf{r}_B - \mathbf{r}_C) =$$
$$\mathbf{a}_{B_2} + a_{B_{32}}(\cos\phi_2 \mathbf{\imath} + \sin\phi_2 \mathbf{\jmath}) + 2\ \boldsymbol{\omega}_3 \times \mathbf{v}_{B_{32}}. \tag{13}$$

Equation (13) represents a vectorial equations with two scalar components on x-axis and y-axis and with two unknowns α_3 and $a_{B_{32}}$

$$-\alpha_3(y_B - y_C) - \omega_3^2(x_B - x_C) = a_{Bx} + a_{B_{32}} \cos\phi_2 - 2\omega_3 v_{B_{32}} \sin\phi_2,$$
$$\alpha_3(x_B - x_C) - \omega_3^2(y_B - y_C) = a_{By} + a_{B_{32}} \sin\phi_2 + 2\omega_3 v_{B_{32}} \cos\phi_2,$$

or

$$-\alpha_3(0.05) - (-7.731)^2(0.086 - 0.050) =$$
$$-2.374 + a_{B_{32}} \cos 53.794° - 2(-7.731)(0.211)\sin 53.794°,$$
$$\alpha_3(0.086 - 0.050) - (-7.731)^2(0.050) =$$
$$-1.370 + a_{B_{32}} \sin 53.794° + 2(-7.731)(0.211)\cos 53.794°.$$

It results

$$\alpha_3 = \alpha_2 = -34.865\ \text{rad/s}^2 \ \text{ and } \ a_{B_{32}} = -1.195\ \ \text{m/s}^2.$$

The velocity of D_3 is

$$\mathbf{v}_{D_3} = \mathbf{v}_{D_4} = \mathbf{v}_C + \boldsymbol{\omega}_3 \times \mathbf{r}_{CD} = \boldsymbol{\omega}_3 \times (\mathbf{r}_D - \mathbf{r}_C) =$$
$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \omega_3 \\ x_D - x_C & y_D - y_C & 0 \end{vmatrix} = \begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & -7.731 \\ 0.138 - 0.050 & 0.121 & 0 \end{vmatrix} =$$
$$0.386\mathbf{\imath} - 0.282\mathbf{\jmath} \text{ m/s}.$$

The acceleration of D_3 is

$$\mathbf{a}_{D_3} = \mathbf{a}_{D_4} = \mathbf{a}_C + \boldsymbol{\alpha}_3 \times \mathbf{r}_{CD} - \omega_3^2 \mathbf{r}_{CD} = \boldsymbol{\alpha}_3 \times (\mathbf{r}_D - \mathbf{r}_C) - \omega_3^2 (\mathbf{r}_D - \mathbf{r}_C) =$$
$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & \alpha_3 \\ x_D - x_C & y_D - y_C & 0 \end{vmatrix} - \omega_3^2 \left[(x_D - x_C)\mathbf{\imath} + (y_D - y_C)\mathbf{\jmath} \right] =$$
$$\begin{vmatrix} \mathbf{\imath} & \mathbf{\jmath} & \mathbf{k} \\ 0 & 0 & -34.865 \\ 0.138 - 0.050 & 0.121 & 0 \end{vmatrix} -$$
$$(-7.731)^2 \left[(0.138 - 0.050)\mathbf{\imath} + (0.121)\mathbf{\jmath} \right] =$$
$$-1.076\mathbf{\imath} - 10.324\mathbf{\jmath} \text{ m/s}^2.$$



Figure 1

```
Apply[Clear, Names["Global`*"]];
Off[General::spell];
Off[General::spell1];

(*Input data*)
AB = 0.1;
AC = 0.05;
CD = 0.15;
φ1 = Pi / 6.;
n = -50.; (*rpm*)
omega = n Pi / 30.; (*rad/s*)

(*Position analysis*)

xA = yA = 0;

xC = AC;
yC = 0;
rC = {xC, yC, 0};

xB = AB Cos[φ1];
yB = AB Sin[φ1];
rB = {xB, yB, 0};

Print["rB = ", rB, " m"];
Print["rC = ", rC, " m"];

mBC = (yB - yC) / (xB - xC);
φ3 = φ2 = ArcTan[mBC];
Print["φ3= ", φ3, " rad = ", φ3 * 180 / N[Pi], " deg "];

xD = xC + CD Cos[φ3];
yD = yC + CD Sin[φ3];
rD = {xD, yD, 0};
Print["rD = ", rD, " m"];

markers = Table[{Point[{xA, yA}], Point[{xB, yB}], Point[{xC, yC}], Point[{xD, yD}]}];

name = Table[{Text["A", {0, 0}, {-1, 1}], Text["B", {xB, yB}, {0, -1}],
    Text["C", {xC, yC}, {1, -1}], Text["D", {xD, yD}, {0, -1}]}];

graph = Graphics[{{RGBColor[1, 0, 0], Line[{{xA, yA}, {xB, yB}}]}, {RGBColor[0, 1, 0],
     Line[{{xC, yC}, {xD, yD}}]}, {RGBColor[1, 1, 1], PointSize[0.01], markers}, {name}}];

Show[Graphics[graph], PlotRange → {{-.05, .2}, {-.05, .15}},
  Frame → True, AxesOrigin → {xA, yA}, FrameLabel → {"x", "y"},
  Axes → {True, True}, AspectRatio → Automatic];

(*Angular velocity and acceleration of the link 1*)
ω = ω1 = {0, 0, omega};
α = α1 = D[ω, t];
Print["ω = ω1 = ", ω, " rad/s"];
Print["α = α1 = ", α, " rad/s^2"];

vB = vB1 = vB2 = Cross[ω1, rB];
Print["vB = vB1 = ω1 x rB = ", vB, " m/s"];
Print["vB1 = vB2 "];
```

```
aB = aB1 = aB2 = Cross[α1, rB] - ω1.ω1 rB;
Print["aB = aB1 = α1 x rB - ω1^2 rB = ", aB, " m/s^2"];
Print["aB1 = aB2 "];

ω3u = {0, 0, ω3z};
Print["ω3 = ω2 = ", ω3u];

vB3u = Cross[ω3u, rB - rC];
Print["vB3 = vC + ω3 x (rB-rC)"];

vB32u = {v32 Cos[ϕ2], v32 Sin[ϕ2], 0};
Print["vB32={vB32 Cos[ϕ2],vB32 Sin[ϕ2],0}"];

Print["vB3 = vB2 + vB32 => ω3z, vB32"];

eqvB = vB3u - vB2 - vB32u;
solutionvB = Solve[{eqvB[[1]] == 0, eqvB[[2]] == 0}, {ω3z, v32}];
ω3 = ω2 = ω3u /. solutionvB[[1]];
vB32 = vB32u /. solutionvB[[1]];
Print["ω3 = ω2 = ", ω3, " rad/s"];
Print["vB32 = ", v32 /. solutionvB[[1]], " m/s"];
Print["vB32v = ", vB32, " m/s"];

vB3 = vB3u /. solutionvB[[1]];
Print["vB3 = ", vB3, " m/s"];

vD3 = vD = Cross[ω3, rD - rC];
Print["vD3 = vD = vC + ω3 x (rD-rC) = ", vD3, " m/s"];


(*--α3--*)
α3u = {0, 0, α3z};
Print["α3 = α2 = ", α3u];

aB3u = Cross[α3u, rB - rC] - ω3.ω3 (rB - rC);
Print["aB3 = aC + α3 x (rB-rC) - ω3.ω3(rB-rC)"];

aB32u = {a32 Cos[ϕ2], a32 Sin[ϕ2], 0};
Print["aB32={aB32 Cos[ϕ2],aB32 Sin[ϕ2],0}"];

aB32cor = 2 Cross[ω3, vB32];
Print["aB32cor = 2 ω3 x vB32 = ", aB32cor, " m/s^2"];

Print["aB3 = aB2 + aB32 + aB32cor => α3z, aB32"];

eqaB = aB3u - aB2 - aB32u - aB32cor;
solutionaB = Solve[{eqaB[[1]] == 0, eqaB[[2]] == 0}, {α3z, a32}];
α3 = α2 = α3u /. solutionaB[[1]];
aB32 = aB32u /. solutionaB[[1]];
Print["α3 = α2 = ", α3, " rad/s^2"];
Print["aB32 = ", a32 /. solutionaB[[1]], " m/s^2"];
Print["aB32v = ", aB32, " m/s^2"];

aB3 = aB3u /. solutionaB[[1]];
Print["aB3 = ", aB3, " m/s^2"];

aD3 = aD = Cross[α3, rD - rC] - ω3.ω3 (rD - rC);
```

```
Print["aD3 = aD = aC + α3 x (rD-rC) - ω3.ω3(rD-rC)"];
Print["aD3 = ", aD3, " m/s^2"];
```

rB = {0.0866025, 0.05, 0} m

rC = {0.05, 0, 0} m

ϕ3= 0.938882 rad = 53.794 deg

rD = {0.138604, 0.121035, 0} m



ω = ω1 = {0, 0, -5.23599} rad/s

α = α1 = {0, 0, 0} rad/s^2

vB = vB1 = ω1 x rB = {0.261799, -0.45345, 0.} m/s

vB1 = vB2

aB = aB1 = α1 x rB - ω1^2 rB = {-2.37426, -1.37078, 0.} m/s^2

aB1 = aB2

ω3 = ω2 = {0, 0, ω3z}

vB3 = vC + ω3 x (rB-rC)

vB32={vB32 Cos[ϕ2],vB32 Sin[ϕ2],0}

vB3 = vB2 + vB32 => ω3z, vB32

ω3 = ω2 = {0, 0, -7.7316} rad/s

vB32 = 0.211245 m/s

vB32v = {0.124781, 0.170454, 0} m/s

vB3 = {0.38658, -0.282996, 0.} m/s

vD3 = vD = vC + ω3 x (rD-rC) = {0.935792, -0.685048, 0.} m/s

α3 = α2 = {0, 0, α3z}

```
aB3 = aC + α3 x (rB-rC) - ω3.ω3(rB-rC)

aB32={aB32 Cos[ϕ2],aB32 Sin[ϕ2],0}

aB32cor = 2 ω3 x vB32 = {2.63576, -1.92951, 0.} m/s^2

aB3 = aB2 + aB32 + aB32cor => α3z, aB32

α3 = α2 = {0, 0, -34.8653} rad/s^2

aB32 = -1.19563 m/s^2

aB32v = {-0.706249, -0.964755, 0} m/s^2

aB3 = {-0.444749, -4.26504, 0.} m/s^2

aD3 = aD = aC + α3 x (rD-rC) - ω3.ω3(rD-rC)

aD3 = {-1.0766, -10.3244, 0.} m/s^2
```